

## C3M9

### Double Integrals in Polar Coordinates

There are regions in the plane that are not easily used as domains of iterated integrals in rectangular coordinates. Sometimes switching to an integral in polar coordinates makes a difficult problem much easier. We will begin with a simple example and show how to make the transition. There are certain *Do's* and *Don't's* that we will point out. First, **ALWAYS** begin by writing the variables of integration in the limits of the **rectangular** integral, such as “ $x =$ ” and “ $y =$ ”. Second, **ALWAYS** draw a sketch of the domain of the integral after the first step and think about how you would draw that same sketch using polar coordinates.


**Example 1:** Evaluate the integral 
$$\int_1^2 \int_0^{\sqrt{4-x^2}} y \, dy \, dx$$
 using polar coordinates.

First Step :

$$\int_1^2 \int_0^{\sqrt{4-x^2}} y \, dy \, dx \quad \longrightarrow \quad \begin{array}{l} x=2 \quad y= \sqrt{4-x^2} \\ x=1 \quad y=0 \end{array} \int y \, dy \, dx$$


Let's use Maple to plot the domain. We will see that we have a portion of the circle of radius 2 in the first quadrant. A fourth plot is used to put the axes in their proper place. The output is to the right below.

```
> with(student):      with(plots):
> A:=plot([x,sqrt(4-x^2),x=1..2]):
> B:=plot([1,y,y=0..sqrt(3)]):
> C:=plot([x,0,x=1..2]):
> E:=plot([0,y,y=0..1]):
> display(A,B,C,E);
```



We are providing more details here than are really necessary, but maybe someone will benefit from it. Now we will plot the same region using `polarplot`. The output is above on the right. We must use the fact that a vertical line  $x = a$  in polar coordinates is represented by  $r = a \sec \theta$ . If  $a = 0$  then we use  $\theta = \pi/2$ .

```
> A1:=polarplot([2,t,t=0..Pi/3]):
> B1:=polarplot([sec(t),t,t=0..Pi/3]):
> C1:=polarplot([r,0,r=1..2]):
> angleline:=polarplot([r,Pi/3,r=0..2.2],color=blue):
> display(A1,B1,C1,E,angleline);
```



We included the plot `angleline` to show the last angle needed to plot the figure. Now we are ready to set up our integral. **NEVER** simply substitute  $x = r \cos \theta$  and  $y = r \sin \theta$  into the limits of the rectangular integral. **ALWAYS** make those substitutions into the integrand  $f(x, y)$  and multiply the result by  $r$  to form  $f(r \cos \theta, r \sin \theta) r \, dr \, d\theta$ . Look at the figure and ask yourself, “How do I draw this figure in polar coordinates?” “Which functions of  $\theta$  does  $r$  range between?” And “For which values does  $\theta$  first touch the

figure and last touch the figure?" The process looks like

$$\begin{array}{ccc} x=b & y=h(x) & \\ x=a & y=g(x) & \end{array} f(x, y) dy dx \longrightarrow \begin{array}{ccc} \theta=\theta_2 & r=v(\theta) & \\ \theta=\theta_1 & r=u(\theta) & \end{array} f(r \cos(\theta), r \sin(\theta)) r dr d\theta$$

Think of a line emanating from the origin and extending out at about 30°. It will first touch the figure on the vertical line and exit the figure at the circle. And, this will remain true as you place that line along the  $x$ -axis and rotate it up to the point where the line and arc intersect. Using Maple to substitute into the integrand:

```
> f:=(x,y)->y;
                                f := (x, y) -> y
> grand:=simplify(f(r*cos(t),r*sin(t)),symbolic);
                                grand := r sin(t)
```

The polar integral becomes


$$\int_{\theta=0}^{\theta=\pi/3} \int_{r=\sec(\theta)}^{r=2} r \sin(\theta) r dr d\theta = \int_0^{\pi/3} \int_{\sec(\theta)}^2 r^2 \sin(\theta) dr d\theta$$

```
> answer1:=Doubleint(grand*r,r=sec(t)..2,t=0..Pi/3);
                                answer1 := \int_0^{\pi/3} \int_{\sec(t)}^2 r^2 \sin(t) dr dt
> answer2:=value(answer1);
                                answer2 := \frac{5}{6}
```

**Example 2:** Find the integral of  $f(x, y) = e^{x^2+y^2}$  over the annular region in the first quadrant between the circles  $x^2 + y^2 = 1$  and  $x^2 + y^2 = 4$ .

It is easiest to graph this using `polarplot`.

```
> with(student): with(plots):
> P1:=polarplot([2,t,t=0..Pi/2]):
> P2:=polarplot([1,t,t=0..Pi/2]):
> display(P1,P2);
```



It is easy to see that this integral cannot be done as one iterated integral in rectangular coordinates.

```
> f:=(x,y)->exp(x^2+y^2);
                                f := (x, y) -> e^{(x^2+y^2)}
> grand:=simplify(f(r*cos(t),r*sin(t)),symbolic);
                                grand := e^{(r^2)}
> Polint:=Doubleint(grand*r,r=1..2,t=0..Pi/2);
                                Polint := \int_0^{1/2 \pi} \int_1^2 e^{(r^2)} r dr dt
> Polint:=value(Polint);
                                Polint := \frac{1}{4} \pi e^4 - \frac{1}{4} \pi e
```

In addition to problems with the region, there is no known antiderivative for  $e^{x^2}$ . The presence of the extra  $r$  in the polar integrals makes such problems a straightforward substitution,  $u = r^2$ .

**C3M9 Problems** Use Maple and polar coordinates to evaluate the given integrals.

1.  $\int_{-2}^2 \int_0^{\sqrt{4-x^2}} e^{-x^2-y^2} dy dx$

2.  $\int_S \frac{x^2}{x^2+y^2} dA$ ,  $S$  is the annular region between  $x^2+y^2=1$  and  $x^2+y^2=3$  with  $y \geq 0$ .

3.  $\int_R \sqrt{x^2+y^2} dA$ ,  $R$  is the triangle with vertices  $(0,0)$ ,  $(3,0)$ , and  $(3,3)$ .

**Challenge!**  $\int_0^2 \int_{-\sqrt{2y-y^2}}^{\sqrt{2y-y^2}} x dx dy$